

wxMaxima pour l'Analyse Volume II

Auteur : Zachary Hannan, Solano Community College
Les corrections sont réalisées par Michel Gosse

Exercices du Chapitre 4 et leur corrigé

Présentation

Le manuel wxMaxima for Calculus Volume II, dont l'auteur est Zachary Hannan, propose l'étude de nombreux domaines de l'analyse à l'aide du logiciel Maxima et de son interface graphique wxMaxima. A l'occasion de la traduction française de ce manuel, la correction des exercices de chaque chapitre est proposée. Ce document reprend les exercices du manuel et leur correction à l'aide de wxMaxima.

La dernière version de ce document, ainsi que les fichiers source pour wxMaxima, sont téléchargeables sur le site <https://maxima-french-doc.fr/>, à la rubrique Documentation.

Enoncé des exercices

1. Montrer que $y(x) = c \cdot e^{x^2/2} + 3$ est la solution générale de l'EDO $\frac{dy}{dx} = xy - 3x$. Utiliser `makelist` pour générer 20 membres de la famille de solutions vérifiant l'EDO. Tracer la famille de solutions en gris ainsi que la solution particulière vérifiant $y(0) = 5$ en rouge.
2. Montrez que $y_1(x) = \sin(\omega x)$ et $y_2(x) = \cos(\omega x)$ sont toutes deux solutions de l'équation différentielle ordinaire (EDO) du second ordre $y''(x) = -\omega^2 y(x)$ (utilisez `%omega` dans wxMaxima). Quelle est la solution générale de cette EDO ? Montrez que $y(x) = A \cdot \cos(\omega x + \phi)$ est une solution générale alternative (avec les constantes arbitraires A et ϕ).
3. Trouvez la solution particulière de $y''(x) = -5y(x)$ soumise aux conditions aux limites $y(0) = 2$ et $y(3,1) = -0,2$. Plutôt que d'utiliser `bc2`, utilisez `subst` et `solve` pour trouver « manuellement » la solution particulière sous deux formes différentes, en utilisant les différentes expressions de la solution générale discutées dans l'exercice précédent. Utilisez `trigexpand` pour montrer que les deux solutions particulières sont équivalentes. Enfin, tracez la solution particulière en incluant les points donnés par les conditions aux limites.
4. Montrez que $c_1 e^{kx} + c_2 e^{-kx}$ est la solution générale de $y''(x) = k^2 \cdot y(x)$. Cette solution générale ressemble fortement à une paire de fonctions « spéciales » — quelles sont-elles ? Montrez que chacune de ces fonctions spéciales vérifie également l'EDO, puis exprimez la solution générale à l'aide de ces fonctions.
5. Utilisez la méthode de séparation des variables pour trouver la solution générale de $y'(x) = -y/x$. Tracez la famille de courbes de la solution générale en utilisant `makelist`.
6. Utilisez la méthode de séparation des variables pour trouver la solution générale de $y'(x) = +y/x$. Tracez la famille de courbes de la solution générale en utilisant `makelist`.
7. Utilisez la méthode de séparation des variables pour trouver la solution particulière de

$$\frac{dy}{dx} = \frac{x^2 \sqrt{x^2 + 3}}{(1 - y^2)^{3/2}}$$

vérifiant la condition initiale $y(3) = 0,5$. Tracez cette solution particulière en vérifiant qu'elle passe par le point $(3, 0,5)$.

8. Utilisez `ode2` pour trouver la solution générale de $y'(x) = -\frac{1}{2}y \cdot \tan x$. Tracez la famille de solutions pour 20 valeurs de `%c` dans la solution générale.
9. Utilisez `ode2` pour trouver la solution générale de $y'(x) = -\sin(y) \cdot \cos(x)$. Tracez la famille de solutions pour 20 valeurs de `%c` dans la solution générale.
10. Tracer un champ de directions pour l'équation différentielle $y'(x) = \frac{\sin^2 x}{y}$, et y inclure un tracé de la solution particulière passant par le point $(0, 1)$.
11. Utiliser la méthode d'Euler avec $n = 100$ pour tracer la solution particulière de l'exercice précédent sur $[0, 5]$.
12. Tracer le champ de directions pour $y'(x) = -\frac{x}{y}$, en incluant les solutions particulières passant par $(0, 1)$, $(0, 2)$ et $(0, 3)$.
13. Pour l'équation différentielle de l'exercice précédent, utiliser la méthode d'Euler avec $n = 100$ pour tracer la solution particulière passant par $(0, 3)$.
14. Un problème classique lié aux équations différentielles consiste à trouver un ensemble de *trajectoires orthogonales* à une famille donnée de courbes. Les trajectoires orthogonales sont des courbes qui coupent la famille donnée à angle droit en chaque point. Considérons la famille de courbes $y = -\frac{C}{x}$. Pour trouver ses trajectoires orthogonales, nous devons d'abord déterminer la pente en tout point, pour toute valeur de C ; c'est-à-dire calculer $\frac{dy}{dx}$ en fonction de C . Dans ce cas, la dérivée est triviale : $\frac{dy}{dx} = \frac{C}{x^2}$.

Pour que les courbes se coupent à angle droit, l'ensemble des trajectoires orthogonales doit avoir en chaque point la pente opposée inverse. Autrement dit, les trajectoires orthogonales vérifient l'équation différentielle $\frac{dy}{dx} = -\frac{x^2}{C}$ (remarquons que $C = 0$ doit être exclu à ce stade). Pour résoudre cette équation différentielle, remplacez C par son expression en fonction de x et y , puis séparez les variables et intégrez (alternativement, vous pouvez utiliser `ode2`). Vous obtiendrez une nouvelle famille de courbes (avec une nouvelle constante arbitraire D) qui doit couper la famille originale à angle droit. Enfin, utilisez `makelist` pour générer des tracés des deux familles de courbes pour $C = -10, -9, \dots, 10$ (en excluant $C = 0$) et $D = -10, -9, \dots, 10$.

15. La *croissance logistique* se produit lorsqu'une population croît presque de manière exponentielle lorsqu'elle est petite, puis se stabilise vers une « capacité de charge » lorsqu'elle devient suffisamment grande. Le modèle de croissance logistique est une équation différentielle

$$\frac{dN}{dt} = kN \cdot (L - N),$$

où N est la taille de la population et L la capacité de charge. On voit de manière évidente que la croissance est presque exponentielle lorsque N est petit :

$$\frac{dN}{dt} \approx (kL)N,$$

et que le taux de croissance tend vers zéro lorsque $N \rightarrow L$. De plus, on observe que le taux de croissance devient négatif si $N > L$; autrement dit, la population diminue si elle dépasse la capacité de charge du milieu.

Supposons qu'une population de lapins croît selon un modèle logistique avec $k = 0.312$ et $L = 1000$. On suppose que les unités de temps sont des semaines.

- (a) Tracer la population en fonction du temps en utilisant les conditions initiales $N(0) = 2$ et $N(0) = 5$.
- (b) Combien de temps faut-il à chaque population pour atteindre 80% de la capacité de charge ?
- (c) Supposons que les lapins aient un problème de surpopulation : $N(0) = 2000$. Tracer la population en fonction du temps. Combien de temps faut-il avant que la population diminue à 1200 lapins ?
- (d) Déterminer le comportement de la population si $N(0) = 1000$.

16. La désintégration radioactive est régie par une physique probabiliste – chaque atome d'un isotope instable porte exactement la même probabilité de désintégration par seconde. Par exemple, si un isotope a une probabilité de désintégration de $k = .001$ par atome et par seconde, et que nous avons une collection de 2000 atomes, nous attendons 2 désintégrations dans la seconde suivante. Ce comportement peut être modélisé par une équation différentielle du premier ordre : $\frac{dN}{dt} = -kN$; autrement dit, le taux de désintégration (atomes par seconde) est proportionnel au nombre d'atomes. La solution de cette équation différentielle est facile à deviner – écrivez-la, en veillant à bien inclure une constante arbitraire. Substituez maintenant $t = 0$, et interprétez la constante arbitraire.

Les atomes de carbone-14 ont une probabilité de désintégration par atome et par *année* d'environ $k \approx .000121$.

- En partant de $N(0) = 10^{23}$, produisez un tracé de $N(t)$ sur une période de 20 000 ans.
- Calculez le pourcentage d'atomes initiaux restants après 10 000 ans.
- Calculez le temps nécessaire pour que la moitié des atomes se désintègre (c'est ce qu'on appelle la demi-vie du carbone-14).

Remarque : le C-14 est utilisé pour la *datation radiométrique* des vestiges archéologiques. La technique réelle nécessite de connaître le *pourcentage* de carbone présent sous forme de C-14 au moment de la mort d'un organisme. Ce pourcentage est relativement stable au cours du temps, car le C-14 est produit par un processus prévisible dans la haute atmosphère terrestre. Lorsqu'un organisme est vivant, il incorpore du carbone provenant de son environnement, y compris cette même fraction de C-14. Lorsqu'un organisme meurt, la fraction de C-14 diminue à mesure que le C-14 se désintègre en N-14, tandis que le carbone « ordinaire » C-12 reste stable. La *fraction* de C-14 suit la même courbe de désintégration que la *quantité* de C-14, en très bonne approximation, car le C-14 n'apparaît qu'à l'état de traces (environ une partie par billion).

- Si l'accélération d'un objet est constante, nous pouvons décrire son comportement par une équation différentielle du second ordre, $x''(t) = a$, où a est l'accélération constante. Résoudre cette équation différentielle avec les conditions initiales $x(0) = x_0$ et $v(0) = v_0$ pour obtenir les équations standard du mouvement à accélération constante $x(t)$ et $v(t)$.
- Trouvez les équations du mouvement pour un objet dont l'accélération est $a(t) = te^{-t}$. Utilisez les conditions initiales standard de l'exercice précédent.

Corrigé des exercices

1 Exercice1

(% i1) declare(c,constant)\$

(% i2) y(x):=c*exp(x^2/2)+3;

(%o2) $y(x) := c \exp\left(\frac{x^2}{2}\right) + 3$

(% i3) diff(y(x),x);

(%o3) $\%e^{\frac{x^2}{2}} cx$

(% i4) expand(x*y(x)-3*x);

(%o4) $\%e^{\frac{x^2}{2}} cx$

On prouve bien ainsi que $y(x)$ est la solution générale de l'EDO.

```
(% i5) famille :makelist(explicit(subst(k,c,y(x)),x,-5,5),k,-9,10)$
```

Recherche de la solution particulière :


```
(% i6) constante :solve(y(0)=5,c);  
(%o6) [c=2]
```

```
(% i7) rhs(constante[1]);  
(%o7) 2
```

```
(% i8) y0(x) :=subst(rhs(constante[1]),c,y(x));  
(%o8) y0(x) := subst (rhs (constante_1) , c , y(x))
```

```
(% i9) y0(x);  
(%o9)  $2e^{\frac{x^2}{2}} + 3$ 
```

```
(% i10) wxdraw2d(  
  title= ,  
  grid=true,  
  xaxis=true,  
  yaxis=true,  
  xrange=[-3,3],  
  yrange=[-15,15],  
  color=grey,  
  line_width=1,  
  key= ,  
  famille,  
  color=red,  
  explicit(y0(x),x,-5,5)  
)$
```



```
(%t10)
```

2 Exercice 2

```
(% i1) kill(all)$declar(w, constant)$
```

```
(% i3) y1(x) := sin(ω*x)$y2(x) := cos(ω*x)$
```

```
(% i4) diff(y1(x),x,2)+ω^2*y1(x);  
(%o4) 0
```

```
(% i5) diff(y2(x),x,2)+ω^2*y2(x);  
(%o5) 0
```

On peut utiliser le fait que l'EDO est de degré 2 et que nous avons 2 solutions qui peuvent générer l'ensemble des solutions :

```
(% i7) declare(k1,constant)$declare(k2,constant)$
```

```
(% i8) y(x) := k1*y1(x)+k2*y2(x);  
(%o8) y(x) := k1 y1(x) + k2 y2(x)
```

```
(% i9) diff(y(x),x,2)+ω^2*y(x);  
(%o9) ω2 (k1 sin (xω) + k2 cos (xω)) - k1ω2 sin (xω) - k2ω2 cos (xω)
```

```
(% i10) ratsimp(%);  
(%o10) 0
```

```
(% i12) declare(A,constant)$declare(φ,constant)$
```

```
(% i13) ya(x) := A*cos(ω*x+φ);  
(%o13) ya(x) := A cos (ωx + φ)
```

```
(% i14) diff(ya(x),x,2)+ω^2*ya(x);  
(%o14) 0
```

La formule $\cos(a+b)=\cos(a)\cos(b)-\sin(a)\sin(b)$ permet de démontrer cette équivalence des familles de solution.

3 Exercice 3

Première expression de la solution générale trouvée à l'exercice 2 :

```
(% i15) y(x);  
(%o15) k1 sin (xω) + k2 cos (xω)
```

Comme $w=\sqrt{5}$ dans cet exercice, on a donc :

```
(% i16) y3(x) := subst(sqrt(5),ω,y(x));  
(%o16) y3(x) := subst (√5, ω, y(x))
```

```
(% i17) y3(x);  
(%o17) k1 sin (√5x) + k2 cos (√5x)
```

```
(% i18) constantes :solve([y3(0)=2,y3(3.1)=-0.2],[k1,k2]);
rat : replaced 0.2 by 1/5 = 0.2
```

```
(constantes)  
$$\left[ \left[ k1 = - \left( \frac{10 \cos(3.1\sqrt{5}) + 1}{5 \sin(3.1\sqrt{5})} \right), k2 = 2 \right] \right]$$

```

```
(% i19) rhs(constantes[1][1]);
```

```
(%o19)  
$$- \left( \frac{10 \cos(3.1\sqrt{5}) + 1}{5 \sin(3.1\sqrt{5})} \right)$$

```

```
(% i20) rhs(constantes[1][2]);
```

```
(%o20)  2
```

```
(% i21) define(y4(x), subst([k1 = rhs(constantes[1][1]), k2 = rhs(constantes[1][2])], y3(x)));
```

```
(%o21)  
$$y4(x) := 2 \cos(\sqrt{5}x) - \frac{(10 \cos(3.1\sqrt{5}) + 1) \sin(\sqrt{5}x)}{5 \sin(3.1\sqrt{5})}$$

```

```
(% i22) ya(x);
```

```
(%o22)   $A \cos(x\omega + \phi)$ 
```

```
(% i23) ya2(x) := subst(sqrt(5),omega,ya(x));
```

```
(%o23)   $ya2(x) := \text{subst}(\sqrt{5}, \omega, ya(x))$ 
```

```
(% i24) ya2(x);
```

```
(%o24)   $A \cos(\sqrt{5}x + \phi)$ 
```

```
(% i25) constantes :solve([ya2(0)=2,ya2(3.1)=-0.2],[A,phi]);
```


```
rat : replaced 0.2 by 1/5 = 0.2
```

```
(constantes)  []
```

On constate que Maxima ne sait pas résoudre ce système d'équations transcendentes... On peut développer avec trigexpand, mais dans ce cas on retrouve la forme y3(x) que nous avons déjà traité.

```
(% i26) wxdraw2d(
  grid=true,
  xaxis=true,
  yaxis=true,
  xrange=[-.5,3.5],
  yrange=[-4,4],
  title= Solution particulière avec conditions aux limites ,
  color=black,
  explicit(y4(x),x,0,3.1),
  color=red,
  point_type=7,
  points([[0,y4(0)],[3.1,y4(3.1)]])
);
```

Solution particulière avec conditions aux limites



(%t26)

(%o26)

4 Exercice 4

(% i27) kill(all)\$

(% i3) declare(k,constant)\$declare(c1,constant)\$declare(c2,constant)\$

(% i4) f(x):=c1*exp(k*x)+c2*exp(-k*x)\$

(% i5) diff(f(x),x,2)-k^2*f(x);

(%o5) $-\left(\left(e^{-kx}c2 + e^{kx}c1\right)k^2\right) + e^{-kx}c2k^2 + e^{kx}c1k^2$

(% i6) ratsimp(%);

(%o6) 0

On a bien prouvé que f est solution de cette EDO. On trouve une similitude avec cosh et sinh :

(% i8) y1(x):=cosh(k*x)\$y2(x):=sinh(k*x)\$

(% i9) diff(y1(x),x,2)-k^2*y1(x);

(%o9) 0

(% i10) diff(y2(x),x,2)-k^2*y2(x);

(%o10) 0

L'équation différentielle étant de degré 2 et linéaire, la solution générale s'écrit donc

(% i11) g(x):=c1*y1(x)+c2*y2(x)\$

(% i12) g(x);

(%o12) $c2 \sinh(kx) + c1 \cosh(kx)$

5 Exercice 5

```
(% i13) kill(all)$
```

```
(% i1) edo :dy/dx=-y/x;
```

```
(edo)  $\frac{dy}{dx} = -\left(\frac{y}{x}\right)$ 
```

```
(% i2) edo2 :edo*dx/y;
```

```
(edo2)  $\frac{dy}{y} = -\left(\frac{dx}{x}\right)$ 
```

```
(% i3) declare(c,constant)$
```

On intègre chaque membre en remplaçant dy et dx par 1 pour l'intégration, et en rajoutant une constante c :

```
(% i4) int1 :integrate(subst(1,dy,lhs(edo2)),y,1);
```

```
(int1)  $\log(y)$ 
```

```
(% i5) int2 :integrate(subst(1,dx,rhs(edo2)),x,1)+c;
```

```
(int2)  $c - \log(x)$ 
```

```
(% i6) sol1 :solve(int1=int2,y);
```

```
(sol1)  $\left[ y = \frac{{\%e^c}}{x} \right]$ 
```

```
(% i7) sol2 :subst(k,%e^ c,sol1);
```


```
(sol2)  $\left[ y = \frac{k}{x} \right]$ 
```

```
(% i8) define(g(x,k),rhs(sol2[1]));
```

```
(%o8)  $g(x,k) := \frac{k}{x}$ 
```

```
(% i10) famille1 :makelist(explicit(g(x,k),x,-3,-0.1),k,-6,6)$famille2 :makelist(explicit(g(x,k),x,0.1,3),k,-6,6)$
```

```
(% i11) wxdraw2d(  
  title= ,  
  grid=true,  
  xaxis=true,  
  yaxis=true,  
  xrange=[-3,3],  
  yrange=[-10,10],  
  color=blue,  
  line_width=1,  
  key= ,  
  famille1 ,famille2  
)$
```



(%t11)

6 Exercice 6

(% i12) kill(all)\$

(% i1) edo :dy/dx=y/x;

(edo) $\frac{dy}{dx} = \frac{y}{x}$

(% i2) edo2 :edo*dx/y;

(edo2) $\frac{dy}{y} = \frac{dx}{x}$

(% i3) declare(c,constant)\$

(% i4) int1 :integrate(subst(1,dy,lhs(edo2)),y,1);

(int1) $\log(y)$

(% i5) int2 :integrate(subst(1,dx,rhs(edo2)),x,1)+c;

(int2) $\log(x) + c$

(% i6) sol1 :solve(int1=int2,y);

(sol1) $[y = \%e^c x]$

(% i7) sol2 :subst(k,%e^c,sol1);


(sol2) $[y = kx]$

(% i8) define(g(x,k),rhs(sol2[1]));

(%o8) $g(x, k) := kx$

```
(% i9) famille :makelist(explicit(g(x,k),x,-3,3),k,-6,6)$
```

```
(% i10) wxdraw2d(
title= ,
grid=true,
xaxis=true,
yaxis=true,
xrange=[-3,3],
yrange=[-10,10],
color=blue,
line_width=1,
key= ,
famille
)$
```



```
(%t10)
```

7 Exercice 7

```
(% i11) kill(all)$
```

```
(% i1) edo :dy/dx=x^2*sqrt(x^2+3)/(1-y^2)^(3/2);
```

$$(edo) \quad \frac{dy}{dx} = \frac{x^2 \sqrt{x^2 + 3}}{(1 - y^2)^{\frac{3}{2}}}$$

```
(% i2) edo2 :edo*dx*(1-y^2)^(3/2);
```

$$(edo2) \quad dy(1 - y^2)^{\frac{3}{2}} = dx x^2 \sqrt{x^2 + 3}$$

```
(% i3) declare(c,constant)$
```

```
(% i4) int1 :integrate(subst(1,dy,lhs(edo2)),y,1);
```

$$(int1) \quad \frac{3 \operatorname{asin}(y)}{8} + \frac{y(1 - y^2)^{\frac{3}{2}}}{4} + \frac{3y\sqrt{1 - y^2}}{8}$$

(% i5) int2 :integrate(subst(1,dx,rhs(edo2)),x,1)+c;

$$(int2) -\left(\frac{9 \operatorname{asinh}\left(\frac{x}{\sqrt{3}}\right)}{8}\right) + \frac{x(x^2+3)^{\frac{3}{2}}}{4} - \frac{3x\sqrt{x^2+3}}{8} + c$$

(% i6) edo3 :int1=int2;

$$(edo3) \frac{3 \operatorname{asin}(y)}{8} + \frac{y(1-y^2)^{\frac{3}{2}}}{4} + \frac{3y\sqrt{1-y^2}}{8} = -\left(\frac{9 \operatorname{asinh}\left(\frac{x}{\sqrt{3}}\right)}{8}\right) + \frac{x(x^2+3)^{\frac{3}{2}}}{4} - \frac{3x\sqrt{x^2+3}}{8} + c$$

(% i7) solpar :sublis([x=3,y=0.5],edo3);

$$(solpar) 0.43991918566373545 = -\left(\frac{9 \operatorname{asinh}(\sqrt{3})}{8}\right) + c + \frac{73^{\frac{5}{2}}}{4}$$

(% i9) c1 :solve(solpar,c);c2 :float(solve(solpar,c));

rat : replaced0.43991918566373545by6944482/15785813 = 0.4399191856637349

$$(c1) \left[c = \frac{142072317 \operatorname{asinh}(\sqrt{3}) - 2210013823^{\frac{5}{2}} + 55555856}{126286504} \right] \text{rat : replaced0.43991918566373545by6944482/15785813} = 0.4399191856637349$$

$$(c2) [c = -25.358303399505658]$$

(% i10) c3 :rhs(c2[1]);

$$(c3) -25.358303399505658$$

(% i11) edo4 :subst(c3,c,edo3);

$$(edo4) \frac{3 \operatorname{asin}(y)}{8} + \frac{y(1-y^2)^{\frac{3}{2}}}{4} + \frac{3y\sqrt{1-y^2}}{8} = -\left(\frac{9 \operatorname{asinh}\left(\frac{x}{\sqrt{3}}\right)}{8}\right) + \frac{x(x^2+3)^{\frac{3}{2}}}{4} - \frac{3x\sqrt{x^2+3}}{8} - 25.358303399505658$$

(% i12) wxdraw2d(

title= Solution particulière ,

grid=true,

xaxis=true,

yaxis=true,

xrange=[0,5],

yrange=[-0.99, 0.99],

color=blue,

line_width=2,

key= ,

implicit(edo4, x, 0, 5, y, -0.99, 0.99),

color=red,

point_type=7,


point_size=2,

points([[3, 0.5]])

)\$

rat : replaced1.7320508075688772by37220045/21489003 = 1.732050807568876rat : replaced25.358303399505658by25.358303399505665

Solution particulière



(%t12)

8 Exercice 8

(% i13) kill(all)\$

(% i1) edo1 :='diff(y,x)=-1/2*y*tan(x);

(edo1) $\frac{d}{dx}y = -\left(\frac{\tan(x)y}{2}\right)$

(% i2) solution :ode2(edo1,y,x);

(solution) $y = \%c e^{-\left(\frac{\log(\sec(x))}{2}\right)} \%c$

(% i3) solution1 :radcan(trigsimp(ratsimp(solution)));


(solution1) $y = \%c \sqrt{\cos(x)}$

(% i4) solution2 :rhs(solution1);

(solution2) $\%c \sqrt{\cos(x)}$

(% i5) famille :makelist(explicit(subst(k,%c,solution2),x,-1.5,1.5),k,-10,10)\$

(% i6) wxdraw2d(
 title= ,
 grid=true,
 xaxis=true,
 yaxis=true,
 xrange=[-1.5,1.5],
 yrange=[-10,10],
 color=blue,
 line_width=1,
 key= ,
 famille
)\$



(%t6)

9 Exercice 9

(% i7) kill(all)\$

(% i1) edo1 :='diff(y,x)=-sin(y)*cos(x);

(edo1) $\frac{d}{dx}y = -(\cos(x) \sin(y))$

(% i2) solution :ode2(edo1,y,x);

(solution) $\frac{\log(\cos(y) + 1) - \log(\cos(y) - 1)}{2} = \sin(x) + \%c$

(% i3) solution1 :logcontract(solution);

(solution1) $-\left(\frac{\log\left(\frac{\cos(y) - 1}{\cos(y) + 1}\right)}{2}\right) = \sin(x) + \%c$

Il nous faut transformer manuellement la solution pour obtenir l'expression de y

(% i4) solution3 :-solution1*2;

(solution3) $\log\left(\frac{\cos(y) - 1}{\cos(y) + 1}\right) = 2(-\sin(x) - \%c)$

(% i5) solution4 :exp(lhs(solution3))=exp(rhs(solution3));

(solution4) $\frac{\cos(y) - 1}{\cos(y) + 1} = \%ce^{2(-\sin(x) - \%c)}$

(% i6) solve(solution4,cos(y));

(%o6) $\left[\cos(y) = -\left(\frac{\%ce^{-(2\sin(x)) - 2\%c} + 1}{\%ce^{-(2\sin(x)) - 2\%c} - 1}\right)\right]$

(% i7) solution5 :solve(solution4,y);

solve : using arc - trig function to get a solution. Some solutions will be lost.

(solution5)
$$\left[y = \pi - \arccos \left(\frac{\%e^{-(2 \sin(x)) - 2\%c}}{\%e^{-(2 \sin(x)) - 2\%c} - 1} + \frac{1}{\%e^{-(2 \sin(x)) - 2\%c} - 1} \right) \right]$$


(% i8) solution6 :rhs(solution5[1]);

(solution6)
$$\pi - \arccos \left(\frac{\%e^{-(2 \sin(x)) - 2\%c}}{\%e^{-(2 \sin(x)) - 2\%c} - 1} + \frac{1}{\%e^{-(2 \sin(x)) - 2\%c} - 1} \right)$$

(% i9) famille :makelist(explicit(subst(k,%c,solution6),x,-0.5,1.5),k,-10,10)\$

(% i10) wxdraw2d(
 title= ,
 grid=true,
 xaxis=true,
 yaxis=true,
 xrange=[-1.5,1.5],
 yrange=[-10,10],
 color=blue,
 line_width=1,
 key= ,
 famille
)\$

expt : undefined : 0toanegativeexponent.



(%t10)

De par la présence de arccos, les courbes sont manquantes suite au problème de la perte de solutions quand on utilise arccos pour trouver y en fonction de x. On refait l'équation en séparant les variables

(% i12) lhs : 1/sin(y)\$
 rhs : -cos(x)\$

(% i14) membre1 :integrate(1/sin(y), y);
 membre2 :integrate(-cos(x), x);

(membre1)
$$\frac{\log(\cos(y) - 1)}{2} - \frac{\log(\cos(y) + 1)}{2}$$

(membre2)
$$-\sin(x)$$

(% i15) factor(membre1);

$$(\%o15) \quad - \left(\frac{\log(\cos(y) + 1) - \log(\cos(y) - 1)}{2} \right)$$

(% i16) logcontract(%);

$$(\%o16) \quad \frac{\log\left(\frac{\cos(y) - 1}{\cos(y) + 1}\right)}{2}$$

En séparant les variables, on retrouve la même expression que celle donnée par ode2. Nous transformons la solution générale à l'aide de formules trigonométriques pour obtenir un résultat plus exploitable :

(% i17) solution1;

$$(\%o17) \quad - \left(\frac{\log\left(\frac{\cos(y) - 1}{\cos(y) + 1}\right)}{2} \right) = \sin(x) + \%c$$

formules trigonométriques : $\cos(2a)+1=2\cos(a)^2$ et $\cos(2a)-1=-2\sin(a)^2$

(% i18) solution2 :subst(2*cos(y/2)^2,cos(y)-1,solution1);

$$(solution2) \quad - \left(\frac{\log\left(\frac{2\cos\left(\frac{y}{2}\right)^2}{\cos(y) + 1}\right)}{2} \right) = \sin(x) + \%c$$

(% i19) solution3 :subst(-2*sin(y/2)^2,cos(y)+1,solution2);

$$(solution3) \quad - \left(\frac{\log\left(-\left(\frac{\cos\left(\frac{y}{2}\right)^2}{\sin\left(\frac{y}{2}\right)^2}\right)\right)}{2} \right) = \sin(x) + \%c$$

L'intégration avec le log se fait avec une valeur absolue. En conséquence, l'intégrande est positif, et donc l'intérieur du logarithme est positif, ce qui nous conduit à modifier le signe

(% i20) solution4 :-log(tan(y/2)^2)/2=rhs(solution3);

$$(solution4) \quad -\log\left(\tan\left(\frac{y}{2}\right)\right) = \sin(x) + \%c$$

(% i21) solution5 :-solution4;

$$(solution5) \quad \log\left(\tan\left(\frac{y}{2}\right)\right) = -\sin(x) - \%c$$

(% i22) prov :subst(A,tan(y/2),solution5);

$$(prov) \quad \log(A) = -\sin(x) - \%c$$

(% i23) prov2 :solve(prov,A);

$$(prov2) \quad \left[A = \%e^{-\sin(x) - \%c} \right]$$

(% i24) solution6 :subst(tan(y/2), A, prov2);

$$(solution6) \quad \left[\tan\left(\frac{y}{2}\right) = \%e^{-\sin(x) - \%c} \right]$$


(% i25) solution7 :y=2*atan(rhs(solution6[1]));

$$(solution7) \quad y = 2 \operatorname{atan}\left(\%e^{-\sin(x) - \%c}\right)$$

```
(% i26) solution8 :rhs(solution7);
(solution8) 2 atan( $\%e^{-\sin(x)} - \%c$ )
```

```
(% i27) famille2 :makelist(explicit(subst(k,%c,solution8),x,-7,7),k,-1,7)$
```

```
(% i28) wxdraw2d(
title= ,
grid=true,
xaxis=true,
yaxis=true,
xrange=[-7,7],
yrange=[-1,7],
color=blue,
line_width=1,
key= ,
famille2
)$
```




```
(%t28)
```

10 Exercice 10

```
(% i29) kill(all)$
```

```
(% i1) load(drawdf)$
```

```
(% i2) wxdrawdf(sin(x)^2/y,[x,-10,10],[y,-10,10],[trajectory_at,0,1]);
```



(%t2)


(%o2) 0

11 Exercice 11

```
(% i6) kill(all)$
      DIFF(x,y) :=sin(x)^ 2/y$
      delx(n) :=(5-0)/n$
      X :0$
      Y :1$
      n :100$
      POINTS :[[X,Y]]$

(% i7) (for k :1 thru n do
      (SLOPE :DIFF(X,Y),
      NEXTY :float(Y+SLOPE*delx(n)),
      NEXTX :float(X+delx(n)),
      POINTS : append(POINTS,[[NEXTX,NEXTY]]),
      Y :NEXTY,
      X :NEXTX)
      )$

(% i8) wxdraw2d(
      xrange=[-0.1,5.1],
      yrange=[-1,3],
      title= n=100 Approximation d Euler ,
      points_joined=true,
      color=black,
      points(POINTS)
      );
```



(%t8)


(%o8)

On vérifie bien la cohérence avec le graphique obtenu à la question précédente.

12 Exercice 12

(% i1) `kill(all)$load(drawdf)$`

(% i2) `wxdrawdf(-x/y,[x,-10,10],[y,-10,10],[trajectory_at,0,1],
[trajectory_at,0,2],[trajectory_at,0,3]
);`



(%t2)

(%o2) 0


13 Exercice 13

Sans précision du texte, on se place sur l'intervalle $[0;5]$.

```
(% i6) kill(all)$
      DIFF(x,y) :=-x/y$
      delx(n) :=(5-0)/n$
      X :0$
      Y :3$
      n :100$
      POINTS :[[X,Y]]$

(% i7) (for k :1 thru n do
      (SLOPE :DIFF(X,Y),
      NEXTY :float(Y+SLOPE*delx(n)),
      NEXTX :float(X+delx(n)),
      POINTS : append(POINTS,[[NEXTX,NEXTY]]),
      Y :NEXTY,
      X :NEXTX)
      )$

(% i8) wxdraw2d(
      xrange=[-0.1,5.1],
      yrange=[-1,3],
      title= n=100 Approximation d Euler ,
      points_joined=true,
      color=black,
      points(POINTS)
      );
```



(%t8)

(%o8)

14 Exercice 14

```
(% i9) kill(all)$

(% i1) declare(C,constant)$

(% i2) famillecourbes :y=-C/x;
```

$$(famillecourbes) \quad y = -\left(\frac{C}{x}\right)$$

```
(% i3) solve(famillecourbes,C);
(%o3) [C = -(xy)]
```

```
(% i4) expressionC :rhs(solve(famillecourbes,C)[1]);
(expressionC) -(xy)
```

```
(% i5) edo : 'diff(y,x)=-x^2/C;
(edo)  $\frac{d}{dx}y = -\left(\frac{x^2}{C}\right)$ 
```

```
(% i6) edo2 :subst(expressionC,C,edo);
(edo2)  $\frac{d}{dx}y = \frac{x}{y}$ 
```

```
(% i7) trajectoires :ode2(edo2,y,x);
(trajectoires)  $\frac{y^2}{2} = \frac{x^2}{2} + \%c$ 
```

```
(% i8) trajectoires2 :subst(D,%c,trajectoires);
(trajectoires2)  $\frac{y^2}{2} = \frac{x^2}{2} + D$ 
```


```
(% i9) familletraj :makelist(
implicit(subst(D_val, D, trajectoires2), x, -5, 5, y, -5, 5),
D_val, -10, 10
)$
```

```
(% i11) famille1 :makelist(explicit(subst(k,C,rhs(famillecourbes)),x,-5,-0.1), k, -10,-1)$
famille3 :makelist(explicit(subst(k,C,rhs(famillecourbes)),x,0.1,5), k, -10,-1)$
```

```
(% i13) famille2 :makelist(explicit(subst(k,C,rhs(famillecourbes)),x,-5,-0.1), k, 1,10)$
famille4 :makelist(explicit(subst(k,C,rhs(famillecourbes)),x,0.1,5), k, 1,10)$
```

On trace en rouge la famille de courbes et en bleu les trajectoires orthogonales.

```
(% i14) wxdraw2d(
title= ,
grid=true,
xaxis=true,
yaxis=true,
xrange=[-5,5],
yrange=[-5,5],
color=blue,
line_width=1,
key= ,
familletraj,
color=red,
famille1,famille2,famille3,famille4
)$
```



(%t14)

15 Exercice 15

(% i15) kill(all)\$

(% i1) edo : 'diff(N,t)=0.312*N*(1000-N);

(edo) $\frac{d}{dt}N = 0.312(1000 - N)N$

(% i2) croissance :ode2(edo,N,t);

rat : replaced-312.0by-312/1 = -312.0rat : replaced0.312by39/125 = 0.312rat : replaced-312.0by-312/1 = -312.0rat : replaced0.312by39/125 = 0.312rat : replaced - 312.0by - 312/1 = -312.0rat : replaced0.312by39/125 = 0.312

(croissance) $\frac{\log(N) - \log(N - 1000)}{312} = t + \%c$

On s'aperçoit qu'il manque des valeurs absolues quand on intègre l'équation différentielle (N est positif mais N-1000 doit aussi être positif). On continue avec Maxima en se rappelant qu'il manque une valeur absolue que nous rajouterons après :

(% i3) croissance2 :solve(croissance,N);

(croissance2) $[\log(N) = 312t + \log(N - 1000) + 312\%c]$

Maxima ne sait pas isoler N en fonction de t, nous allons donc le guider :

(% i4) croissance3 :croissance2[1];

(croissance3) $\log(N) = 312t + \log(N - 1000) + 312\%c$

(% i5) croissance4 :croissance3-log(N-1000);

(croissance4) $\log(N) - \log(N - 1000) = 312t + 312\%c$

(% i6) croissance5 :logcontract(croissance4);

(croissance5) $\log\left(\frac{N}{N - 1000}\right) = 312t + 312\%c$

(% i7) membre1 :lhs(croissance5);

(membre1) $\log\left(\frac{N}{N-1000}\right)$

(% i8) croissance6 :exp(membre1)=exp(rhs(croissance5));

(croissance6) $\frac{N}{N-1000} = \%e^{312t + 312\%c}$

(% i9) croissance7 :abs(lhs(croissance6))=rhs(croissance6);

(croissance7) $\frac{|N|}{|N-1000|} = \%e^{312t + 312\%c}$

Pour N(0)=2, on a donc

(% i10) croissance8 :subst(2,N,croissance7);

(croissance8) $\frac{1}{499} = \%e^{312t + 312\%c}$

(% i11) croissance9 :subst(0,t,croissance8);

(croissance9) $\frac{1}{499} = \%e^{312\%c}$

(% i12) croissance10 :log(lhs(croissance9))=log(rhs(croissance9));

(croissance10) $-\log(499) = 312\%c$

(% i13) const1 :solve(croissance10,%c);

(const1) $\left[\%c = -\left(\frac{\log(499)}{312}\right) \right]$

(% i14) const2 :rhs(float(const1)[1]);

(const2) -0.01991219902484461

(% i15) croissance11 :subst(const2,%c,croissance7);

(croissance11) $\frac{|N|}{|N-1000|} = \%e^{312t - 6.212606095751519}$

Dans le cas N(0)=2, on a N<1000 donc nous pouvons enlever la valeur absolue en changeant de signe :

(% i16) croissance12 :N/(1000-N)=rhs(croissance11);

(croissance12) $\frac{N}{1000-N} = \%e^{312t - 6.212606095751519}$

(% i17) croissance13 :solve(croissance12,N);

rat : replaced $-6.212606095751519by -72721437/11705464 = -6.212606095751522$

(croissance13) $\left[N = \frac{1000\%e^{312t - \frac{72721437}{11705464}}}{\%e^{312t - \frac{72721437}{11705464}} + 1} \right]$

Ce qui permet de définir la première solution vérifiant N(0)=2 que nous appelons f1

(% i18) define(f1(t), ev(rhs(croissance13[1]), numer));

(%o18) $f1(t) := \frac{1000\%e^{312t - 6.212606095751522}}{\%e^{312t - 6.212606095751522} + 1}$

Pour N(0)=5, on reprend les calculs pour trouver f2 de manière similaire

(% i19) croissance14 :subst(5,N,croissance7);

(croissance14) $\frac{1}{199} = \%e^{312t + 312\%c}$

(% i20) croissance15 :subst(0,t,croissance14);

(croissance15) $\frac{1}{199} = \%e^{312\%c}$

(% i21) croissance16 :log(lhs(croissance15))=log(rhs(croissance15));

(croissance16) $-\log(199) = 312\%c$

(% i22) const3 :solve(croissance16,%c);

(const3) $\left[\%c = -\left(\frac{\log(199)}{312} \right) \right]$

(% i23) const4 :rhs(float(const3)[1]);

(const4) -0.01696572059206568

(% i24) croissance17 :subst(const4,%c,croissance7);

(croissance17) $\frac{|N|}{|N - 1000|} = \%e^{312t - 5.293304824724492}$

(% i25) croissance18 :N/(1000-N)=rhs(croissance17);

(croissance18) $\frac{N}{1000 - N} = \%e^{312t - 5.293304824724492}$

(% i26) croissance18 :solve(croissance18,N);

rat : replaced -5.293304824724492 by $-46831911/8847386 = -5.2933048247245$

(croissance18) $\left[N = \frac{1000\%e^{312t - \frac{46831911}{8847386}}}{\%e^{312t - \frac{46831911}{8847386}} + 1} \right]$

(% i27) define(f2(t), ev(rhs(croissance18[1]), numer));

(%o27) $f2(t) := \frac{1000\%e^{312t - 5.2933048247245}}{\%e^{312t - 5.2933048247245} + 1}$


On peut vérifier que les conditions initiales sont bien validées :

(% i29) float(f1(0));float(f2(0));

(%o28) 1.999999999999993

(%o29) 4.999999999999961

```
(% i30) wxdraw2d(
  grid=true,
  xaxis=true,
  yaxis=true,
  xrange=[0,0.1],
  yrange=[-5,1500],
  key=f1(t) ,
  color=black,
  line_width=1,
  explicit(f1(t),t,0,0.1),
  color=blue,
  key= f2(t) ,
  explicit(f2(t),t,0,0.1),
  color=red,
  key= ,
  explicit(800,t,0,0.1)
)$
```



(%t30)

Cherchons les valeurs pour lesquelles on atteint 800 (en semaines)

```
(% i32) t80_1 : find_root(f1(t) = 800, t, 0, 0.1);
```

```
t80_2 : find_root(f2(t) = 800, t, 0, 0.1);
```

```
(t80_1) 0.02435545018228017
```

```
(t80_2) 0.02140897174950125
```

Pour la question (c), on a $N > 1000$, donc la valeur absolue du dénominateur devient $N - 1000$. On reprend donc le raisonnement pour définir $f3$:

```
(% i33) croissance7;
```

```
(%o33)  $\frac{|N|}{|N - 1000|} = e^{312t + 312\%c}$ 
```

Pour $N(0)=2000$ on calcule la constante

```
(% i34) croissance19 :subst(2000,N,croissance7);
```

```
(croissance19)  $2 = e^{312t + 312\%c}$ 
```

```
(% i35) croissance20 :subst(0,t,croissance19);
```

```
(croissance20)  $2 = e^{312\%c}$ 
```

```
(% i36) croissance21 :log(lhs(croissance20))=log(rhs(croissance20));
(croissance21)  $\log(2) = 312\%c$ 
```

```
(% i37) const3 :solve(croissance21,%c);
(const3)  $\left[ \%c = \frac{\log(2)}{312} \right]$ 
```

```
(% i38) const4 :rhs(float(const3)[1]);
(const4) 0.0022216255787177734
```

```
(% i39) croissance22 :subst(const4,%c,croissance7);
(croissance22)  $\frac{|N|}{|N - 1000|} = \%e^{312t + 0.6931471805599453}$ 
```

```
(% i40) croissance23 :N/(N-1000)=rhs(croissance22);
(croissance23)  $\frac{N}{N - 1000} = \%e^{312t + 0.6931471805599453}$ 
```


```
(% i41) croissance24 :solve(croissance23,N);
rat : replaced 0.6931471805599453 by 13614799/19642003 = 0.693147180559946
```

```
(croissance24)  $\left[ N = \frac{1000\%e^{\frac{312t + \frac{13614799}{19642003}}}{\%e^{\frac{312t + \frac{13614799}{19642003}} - 1}} \right]$ 
```

```
(% i42) define(f3(t), ev(rhs(croissance24[1]), numer));
(%o42)  $f3(t) := \frac{1000\%e^{312t + 0.693147180559946}}{\%e^{312t + 0.693147180559946} - 1}$ 
```

```
(% i43) f3(0);
(%o43) 1999.9999999999986
```

```
(% i44) wxdraw2d(
  grid=true,
  xaxis=true,
  yaxis=true,
  xrange=[0,0.1],
  yrange=[-5,2200],
  key= f3(t) ,
  color=black,
  line_width=1,
  explicit(f3(t),t,0,0.1),
  color=red,
  key= ,
  explicit(1200,t,0,0.1)
)$
```



(%t44)

```
(% i45) find_root(f3(t)=1200,t,0,0.02);
```

```
(%o45) 0.0035211932329106068
```

La population est bien décroissante et elle atteint 1200 en 0,0035 semaines. Pour la dernière question, si $N(0)=1000$ alors la population est stable à 1000 car l'équation différentielle devient $dN/dt=0$, soit $N(t)=1000$.

16 Exercice 16

```
(% i1) kill(all)$declare(k,constant)$
```

```
(% i2) edo :`diff(N,t)=-k*N;
```

```
(edo)  $\frac{d}{dt}N = -(kN)$ 
```

```
(% i3) atomes :ode2(edo,N,t);
```

```
(atomes)  $N = \%e^{-(kt)}\%c$ 
```

```
(% i4) subst(0,t,atomes);
```

```
(%o4)  $N = \%c$ 
```

La constante arbitraire $\%c$ est donc égale au nombre initial d'atomes.

```
(% i5) atomes2 :subst(0.000121,k,atomes);
```

```
(atomes2)  $N = \%e^{-(1.2110^{-4}t)}\%c$ 
```


```
(% i6) atomes3 :subst(10^23,%c,atomes2);
```

```
(atomes3)  $N = 1000000000000000000000000\%e^{-(1.2110^{-4}t)}$ 
```

```
(% i7) define(N(t),rhs(atomes3));
```

```
(%o7)  $N(t) := 1000000000000000000000000\%e^{-(1.2110^{-4}t)}$ 
```

```
(% i8) wxdraw2d(
      grid=true,
      xaxis=true,
      yaxis=true,
      xrange=[0,20000],
      yrange=[0,10^23],
      key= N(t) ,
      color=black,
      line_width=1,
      explicit(N(t),t,0,20000)
    )$
```



```
(%t8)
```

```
(% i9) N(10000)*100/10^23;
(%o9) 29.81972794298874
```

Il reste 29% d'atomes après 10000 ans

```
(% i10) find_root(N(t)=10^23/2,t,0,20000);
(%o10) 5728.489095536738
```

Il faut donc environ 5728 années pour la demi-vie.

17 Exercice 17

```
(% i3) kill(all)$declare(a,constant)$declare(x0,constant)$declare(v0,constant)$
```

```
(% i4) edo : 'diff(x,t,2)=a;
```

```
(edo)  $\frac{d^2}{dt^2}x = a$ 
```

```
(% i5) ode2(edo,x,t);
```

```
(%o5)  $x = \frac{at^2}{2} + \%k2t + \%k1$ 
```

On définit la fonction mouvement x(t)

```
(% i6) define(x(t),rhs(ode2(edo,x,t)));
```

$$(\%o6) \quad x(t) := \frac{at^2}{2} + \%k2t + \%k1$$

On définit la fonction vitesse $v(t)$

$$(\%i7) \quad \text{define}(v(t), \text{diff}(x(t), t));$$

$$(\%o7) \quad v(t) := at + \%k2$$

$$(\%i8) \quad v(t);$$

$$(\%o8) \quad at + \%k2$$

$$(\%i9) \quad \text{solutions} : \text{linsolve}([x(0)=x0, v(0)=v0], [\%k1, \%k2]);$$

$$(\text{solutions}) \quad [\%k1 = x0, \%k2 = v0]$$

$$(\%i10) \quad \text{sublis}(\text{solutions}, x(t));$$

$$(\%o10) \quad \frac{at^2}{2} + v0t + x0$$

$$(\%i11) \quad \text{sublis}(\text{solutions}, v(t));$$

$$(\%o11) \quad at + v0$$

On retrouve bien les équations classiques d'un mouvement à accélération constante.

18 Exercice 18

$$(\%i2) \quad \text{kill}(\text{all})\$ \text{declare}(x0, \text{constant})\$ \text{declare}(v0, \text{constant})\$$$

$$(\%i3) \quad \text{edo} : \text{diff}(x, t, 2) = t * \exp(-t);$$

$$(\text{edo}) \quad \frac{d^2}{dt^2} x = \%e^{-t} t$$

$$(\%i4) \quad \text{ode2}(\text{edo}, x, t);$$

$$(\%o4) \quad x = \%e^{-t} (t + 2) + \%k2t + \%k1$$

$$(\%i5) \quad \text{define}(x(t), \text{rhs}(\text{ode2}(\text{edo}, x, t)));$$

$$(\%o5) \quad x(t) := \%e^{-t} (t + 2) + \%k2t + \%k1$$

$$(\%i6) \quad \text{define}(v(t), \text{diff}(x(t), t));$$

$$(\%o6) \quad v(t) := -(\%e^{-t} (t + 2)) + \%k2 + \%e^{-t}$$

$$(\%i7) \quad \text{solutions} : \text{linsolve}([x(0)=x0, v(0)=v0], [\%k1, \%k2]);$$

$$(\text{solutions}) \quad [\%k1 = x0 - 2, \%k2 = v0 + 1]$$

$$(\%i8) \quad \text{sublis}(\text{solutions}, x(t));$$

$$(\%o8) \quad \%e^{-t} (t + 2) + (v0 + 1)t + x0 - 2$$

$$(\%i9) \quad \text{sublis}(\text{solutions}, v(t));$$

$$(\%o9) \quad -(\%e^{-t} (t + 2)) + v0 + \%e^{-t} + 1$$